

EEL7030 Microprocessadores – Laboratório 5

Prof. Raimes Moraes

Dispositivos de Entrada e Saída - Interface com Cristal líquido

Compile e execute o programa abaixo para escrever mensagem no LCD. Para que o exemplo seja executado a uma maior velocidade, colocar *update frequency* do EDSIM51 em 50000.

```
;Programa      ESCRIVE_MSG.asm

CS             EQU    P0.7
EN             EQU    P1.2
RS_0           EQU    0
RS_1           EQU    00001000b
Atraso        EQU    50H

                ORG    0H

                CLR    CS      ; INIBE DECODIFICADOR DOS DISPLAYS DE 7 SEGMENTOS
                CALL   INITDSP  ; ROTINA QUE CONFIGURA CONTROLADOR LCD

;ESCREVE MENSAGEM

                MOV     DPTR,#MENS
                MOV     R2,#RS_1      ; SETB P1.3 (RS) - ENVIO DE DADO para LCD

                CALL    WRT
                JMP     $

MENS: DB 13,"EEL7030 - LCD"

DELAY:         MOV     R0,#Atraso
                DJNZ    R0,$
                RET

INITDSP:       ; subrotina para inicializar o display
                ; 001(DL)_NFxx = 0010 1000b (function set)
                ; DL=0: interface 4 bits; N=1: 2 linhas; F=0: caractere 5x8
                ; En (P1.2): 1-> 0 = escreve; RS (P1.3): 0=comando;1=dado

                MOV     P1,#20H      ; FUNCTION SET - high nibble = 0010b -- interface 4 bits
                SETB    EN           ; GERA EN
                CLR     EN

                CALL    DELAY        ; AGUARDA LCD FICAR PRONTO

                ; Nibble alto do Function Set é enviado 2x.

                MOV     R2,#RS_0      ; CLR P1.3 (RS) - comando vai ser enviado para LCD
                MOV     DPTR,#comando
```

```

CALL WRT          ; escreve dados para o LCD
RET
comando: DB 03h,28h,0fh,06h ; nro. de comandos - function set - display on/off - entry mode

WRT:
MOV A,#0          ; END. DO NRO DE COMANDOS/DADOS
MOVC A,@A+DPTR
MOV R6,A          ; R6=NRO DE COMANDOS/DADOS

MOV R1,#1H        ; DESLOCAMENTO DO COMANDO/MENSAGEM INICIAL

LOOP:
MOV A,R1          ; END. DO PRIMEIRO DADO/COMANDO EM A
MOVC A,@A+DPTR
MOV B,A           ; BYTE A SER ESCRITO EM B
ANL A,#0F0H       ; APAGA NIBBLE LS
ORL A,R2 ; R2 DEVE CONTER RS (0: COMANDO; 8: DADO) ou seja valor de P1.3
MOV P1,A          ; ENVIA PARA LCD
SETB EN           ; GERA EN
CLR EN
MOV A,B
SWAP A            ;TROCA NIBBLES MS-LS

ANL A,#0F0H       ; APAGA NIBBLE LS
ORL A,R2          ; SETB P1.3 (RS) se R2 = RS_1;
MOV P1,A          ; ENVIA PARA LCD
SETB EN           ; GERA EN
CLR EN
CALL DELAY        ; AGUARDA LCD ESTAR PRONTO PARA NOVO COMANDO
INC R1            ; R1 APONTA PARA PRÓXIMO COMANDO/DADO
DJNZ R6,LOOP      ; VERIFICA SE ÚLTIMO DADO/COMANDO
RET
END

```

Exercícios:


- 1) Informar no LCD valor especificado pelas chaves conectadas à porta P2; por exemplo, quando as chaves 0 e 1 estiverem pressionadas, mostrar a mensagem "Valor: 3". OBS: particionar a rotina WRT para criar uma segunda subrotina que envie apenas um caractere de dado ou comando para o LCD. Utilizar comando para posicionar o cursor (*Set DDRAM address*) sobre coluna onde o valor lido será atualizado. O valor lido das chaves deve ser convertido no ASCII correspondente a ser enviado para o LCD; sugestão, utilize a subrotina CONVERTE do Laboratório 3, modificando os valores da TABELA para converter o valor lido de P2 para ASCII.

- 2) Fazer programa que rotacione a mensagem “EEL7030 – 8051” da direita para a esquerda no LCD. OBS: A mensagem deve ser escrita no endereço 0H e no endereço 10H. Utilize comando para deslocar a mensagem ciclicamente (*display shift*). Empregar a rotina de atraso apresentada a seguir entre os comandos para rotacionar a mensagem. Ver figura abaixo para melhor compreensão.

```

DELAY2:
VOL:      MOV    R1, #0F0H
          MOV    R0, #0FFH
          DJNZ   R0, $
          DJNZ   R1, VOL
          RET
          END

```



- 3) Modificar programa anterior para rotacionar a mensagem “EEL7030 – 8051” da direita para a esquerda no LCD se a chave P2.7 estiver aberta; da esquerda para a direita se a chave P2.7 estiver fechada. OBS: A mensagem deve ser escrita no endereço 0H e no endereço

10H. Utilize comando para deslocar a mensagem ciclicamente em uma ou outra direção.

- 4) Fazer um programa para o EDSIM51 que mostre no cristal líquido (LCD) o valor de tecla pressionada no keypad (modo AND Enabled, Pulse). Faça a identificação da tecla pressionada quando a interrupção externa 1 por borda for identificada. Baseie-se nas soluções dos exercícios 3.8 e 4.1. Obs: Os 4 pinos menos significativos da porta P0 (P0.3 a P0.0) devem ser inicializados com '0' (`mov p0,#01110000b`) para que a interrupção externa 1 seja gerada quando uma tecla for pressionada. Neste exemplo, iniba interrupções (EA) no início da execução da rotina KEYPAD; depois de executada, limpar pendência (IE1) e voltar a habilitar as interrupções. Isto é necessário, pois a execução da rotina KEYPAD gera transição de '1' para '0' no pino /INT1, solicitando nova interrupção. Após execução da rotina KEYPAD, executar novamente a instrução `mov p0,#01110000b`. Modificar a TABELA da subrotina CONVERTE do Laboratório 3 para:

TABELA: DB 31H, 32H, 33H, 34H, 35H, 36H, 37H, 38H, 39H, 41H, 30H, 43H, 45H